
Subject: Backlogged question: Drawing vector fields with same scaling in New Graphics

Posted by [tianhuachengyue](#) on Tue, 03 Nov 2015 02:02:06 GMT

[View Forum Message](#) <> [Reply to Message](#)

This is an unsolved problem about the `vector()` function in IDL. Gordon used to post this question on the site:

<http://compgroups.net/comp.lang.idl-pvwave/drawing-vector-fields-with-new-graphics/2090071>

But actually it wasn't quite solved:

Is there a way to plot (using New Graphics) two different vector fields on the same set of axes such that the vector fields have the same scaling? Below is a minimal working program. What I want (and sort of expect) is that the `v2` vectors be proportionally scaled with respect to the `v1` vectors. What I get is that the `v2` vectors appear smaller than the `v1` vectors, even though they are clearly the same in magnitude.

I followed the suggestions to set `v1.length_scale = 2` and

`vmag = mean(sqrt(vx^2 + vy^2))`,

then set `v2.length_scale = 2. / vmag` to try to let `v2` have the same scale as `v1`. But clearly it didn't help... Please see my codes below.

Anyone can help me out? Thanks!

PRO test_vector

```
x = [0.,1.,2.]
y = [0.,0.,0.]
```

```
vx = [1.,1.,1.]
vy = [1.,1.,1.]
vmag = mean(sqrt(vx^2 + vy^2))
```

```
v1 = vector(vx, vy, x, y, $
  XTITLE='X', YTITLE='Y', $
  X RANGE=[-1.,4.], Y RANGE=[-1.,4.])
v1.arrow_thick = 2
v1.length_scale = 2
```

```
x = [1.,2.]
y = [1.,1.]
vx = [-1.0,-1.0]
vy = [-1.0,-1.0]
```

```
v2 = vector(vx, vy, x, y, $
  /OVERPLOT, X RANGE=[-1.,4.], Y RANGE=[-1.,4.])
v2.arrow_thick = 2
v2.length_scale = 2. / vmag
```

END

Subject: Re: Backlogged question: Drawing vector fields with same scaling in New Graphics

Posted by chris_torrence@NOSPAM on Mon, 09 Nov 2015 17:20:39 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Monday, November 2, 2015 at 7:02:11 PM UTC-7, tianhuachengyue wrote:

> This is an unsolved problem about the vector() function in IDL. Gordon used to post this question on the site:

> <http://compgroups.net/comp.lang.idl-pvwave/drawing-vector-fields-with-new-graphics/2090071>

> But actually it wasn't quite solved:

> Is there a way to plot (using New Graphics) two different vector fields on the same set of axes such that the vector fields have the same scaling? Below is a minimal working program. What I want (and sort of expect) is that the v2 vectors be proportionally scaled with respect to the v1 vectors. What I get is that the v2 vectors appear smaller than the v1 vectors, even though they are clearly the same in magnitude.

>
> I followed the suggestions to set v1.length_scale = 2 and
> vmag = mean(sqrt(vx^2 + vy^2)),
> then set v2.length_scale = 2. / vmag to try to let v2 have the same scale as v1. But clearly it didn't help... Please see my codes below.
> Anyone can help me out? Thanks!

>
> PRO test_vector
>
> x = [0.,1.,2.]
> y = [0.,0.,0.]
>
> vx = [1.,1.,1.]
> vy = [1.,1.,1.]
> vmag = mean(sqrt(vx^2 + vy^2))
>
> v1 = vector(vx, vy, x, y, \$
> XTITLE='X', YTITLE='Y', \$
> X RANGE=[-1.,4.], Y RANGE=[-1.,4.])
> v1.arrow_thick = 2
> v1.length_scale = 2
>
> x = [1.,2.]
> y = [1.,1.]
> vx = [-1.0,-1.0]
> vy = [-1.0,-1.0]
>
> v2 = vector(vx, vy, x, y, \$

```
> /OVERPLOT, XRANGE=[-1.,4.], YRANGE=[-1.,4.])
> v2.arrow_thick = 2
> v2.length_scale = 2. / vmag
>
> END
>
> Huazeng
```

Hi Huazeng,

I think you need to scale the second vector by the ratio of the magnitudes. Try this example:

```
x1 = [0.,1.,2.]
y1 = [0.,0.,0.]
vx1 = [3.,1.,1.]
vy1 = [3.,1.,1.]
vmag1 = max(sqrt(vx1^2 + vy1^2))

x2 = [0.5,1.5]
y2 = [0.,0.]
vx2 = [2.0,1.0]
vy2 = [2.0,1.0]
vmag2 = max(sqrt(vx2^2 + vy2^2))
print, vmag1, vmag2

v1 = vector(vx1, vy1, x1, y1, $
  XTITLE='X', YTITLE='Y', $
  XRANGE=[-1.,4.], YRANGE=[-1.,4.])
v1.arrow_thick = 2
v1.length_scale = 4

v2 = vector(vx2, vy2, x2, y2, color='red', $
  /OVERPLOT, XRANGE=[-1.,4.], YRANGE=[-1.,4.])
v2.arrow_thick = 2
v2.length_scale = 4*vmag2/vmag1
```

Cheers,
Chris

Subject: Re: Backlogged question: Drawing vector fields with same scaling in New Graphics
Posted by [tianhuachengyue](#) on Wed, 11 Nov 2015 22:15:12 GMT
[View Forum Message](#) <> [Reply to Message](#)

Hi Chris,
I tried what you suggested and now the plots are working perfectly! Thanks a lot.

BTW, any trick behind using $\text{vmag} = \max(\sqrt{v_x^2 + v_y^2})$ instead using $\text{mean}(\sqrt{v_x^2 + v_y^2})$?

It seems to me the vector plot setting has sth to do with the maximum values of the two sets of vectors (v_x and v_y).

Cheers,
Huazeng

Huazeng

--

Subject: Re: Backlogged question: Drawing vector fields with same scaling in New Graphics

Posted by [tianhuachengyue](#) on Wed, 11 Nov 2015 22:41:40 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi Chris,

I tried setting $v2.length_scale = \text{const} * \text{vmag2} / \text{vmag1}$ and now the plots are working perfectly. Thanks for the solution!

BTW, I also tried $\text{vmag} = \text{mean}(\sqrt{v_x^2 + v_y^2})$ but this didn't work correctly. Any trick behind using $\text{vmag} = \max(\sqrt{v_x^2 + v_y^2})$ instead using $\text{mean}(\sqrt{v_x^2 + v_y^2})$? It seemed to me the vector plot setting is related to the maximum value of vector magnitude.

Cheers,
Huazeng

Subject: Re: Backlogged question: Drawing vector fields with same scaling in New Graphics

Posted by [chris_torrence@NOSPAM](#) on Thu, 12 Nov 2015 16:29:35 GMT

[View Forum Message](#) <> [Reply to Message](#)

On Wednesday, November 11, 2015 at 3:41:43 PM UTC-7, tianhuachengyue wrote:

> Hi Chris,

>

> I tried setting $v2.length_scale = \text{const} * \text{vmag2} / \text{vmag1}$ and now the plots are working perfectly. Thanks for the solution!

>

> BTW, I also tried $\text{vmag} = \text{mean}(\sqrt{v_x^2 + v_y^2})$ but this didn't work correctly. Any trick behind using $\text{vmag} = \max(\sqrt{v_x^2 + v_y^2})$ instead using $\text{mean}(\sqrt{v_x^2 + v_y^2})$? It seemed to me the vector plot setting is related to the maximum value of vector magnitude.

>

> Cheers,
> Huazeng

Hi Huazeng,

Yes, you are correct - it is using the maximum magnitude. Using the mean would be a bad idea, as your maximum vectors might be huge but still have a reasonable mean. You can actually see a lot of this code yourself, in `idlitvisvector__define.pro`.

Cheers,
Chris
